```
model ParticleHeatExchanger
  extends Icons.HeatExchanger;
  //Geometry variables
  parameter Integer n = 100 "number of discretizations";
  parameter Types.Length hc_s = 0.006 "solid channel width";
  parameter Types.Length hc_CO2 = 0.001 "CO2 channel width";
  parameter Types.Length t_m = 0.003 "metal thickness";
  parameter Types.Length H = 1 "heat exchanger height";
  parameter Types.Length W = 0.5 "heat exchanger width";
  parameter Integer N_plate = 25 "number of parallel plates";
  parameter Types.Convection h_conv_CO2 = 1000 "CO2 convection
coefficient";
  parameter Types.Convection h_conv_sw = 180 "solid-wall
convection coefficient";
  parameter Real dx = H / (n - 2) "discretization size";
  //Particle variables
  Types.MassFlowRate m_dot_s_in "inlet mass flow rate";
  Types.MassFlowRate m_dot_s_out "outlet mass flow rate";
  Types.Temperature T_s_in "inlet temperature";
  Types.Temperature T_s_out "outlet temperature";
  Types.Temperature T_s[n](each start = 273.15 + 25) "solid
temperature distribution";
  Types.Enthalpy h_s[n] "particle enthalpy";
  Types.SpecificHeatCapacity cp_s "particle specific heat";
  Types.Density rho_s "particle density";
  parameter Real phi_s = 0.6 "solid volume fraction";
  Types.Velocity v_s "solid velocity";
  //CO2 variables
  Types.MassFlowRate m_dot_CO2_in "inlet mass flow rate";
  Types.MassFlowRate m_dot_CO2_out "outlet mass flow rate";
  Types.Temperature T_CO2_in "inlet temperature";
  Types.Temperature T_CO2_out "outlet temperature";
  Types.Temperature T_CO2[n](each start = 273.15 + 25) "CO2
temperature distribution";
  Types.Enthalpy h_CO2[n] "CO2 enthalpy";
  Types.SpecificHeatCapacity cp_CO2 "CO2 specific heat";
  Types.Density rho_CO2 "CO2 density";
  Types.Velocity v_CO2 "CO2 velocity";
  Types.Heat Q_CO2 "sCO2 heat addition";
  //Heat transfer surface
  Types.Temperature T_m[n](each start = 273.15 + 25) "metal
temperature distribution";
  Types.SpecificHeatCapacity cp_m "metal specific heat";
  Types.Density rho_m "metal density";
  //Connections
  Interfaces.ParticleFlow ParticleInlet annotation(
```

```
    Placement(visible = true, transformation(origin = {0, 100},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {0, 100}, extent = {{-10, -10}, {10,
10}}, rotation = 0)));
  Interfaces.ParticleFlow ParticleOutlet annotation(
    Placement(visible = true, transformation(origin = {0, -100},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {0, -100}, extent = {{-10, -10},
{10, 10}}, rotation = 0)));
  Interfaces.CO2Flow CO2Inlet annotation(
    Placement(visible = true, transformation(origin = {-98, 0},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {-98, 0}, extent = {{-10, -10}, {10,
10}}, rotation = 0)));
  Interfaces.CO2Flow CO2Outlet annotation(
    Placement(visible = true, transformation(origin = {100, 0},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {100, 0}, extent = {{-10, -10}, {10,
10}}, rotation = 0)));
  Modelica.Blocks.Interfaces.RealOutput CO2OutletTemperature
annotation(
    Placement(visible = true, transformation(origin = {98, -22},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {98, -22}, extent = {{-10, -10},
{10, 10}}, rotation = 0)));
  Modelica.Blocks.Interfaces.RealOutput
ParticleOutletTemperature annotation(
    Placement(visible = true, transformation(origin = {-24, -
96}, extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {-24, -96}, extent = {{-10, -10},
{10, 10}}, rotation = -90)));
  Modelica.Blocks.Interfaces.RealInput m_dot_CO2 annotation(
    Placement(visible = true, transformation(origin = {-24, -
96}, extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {-26, 96}, extent = {{-10, -10},
{10, 10}}, rotation = -90)));
equation
//Connections
  m_dot_s_in = ParticleInlet.m_dot;
  m_dot_s_out = -ParticleOutlet.m_dot;
  T_s_in = ParticleInlet.T;
  T_s_out = ParticleOutlet.T;
  m_dot_CO2_in = CO2Inlet.m_dot;
  m_dot_CO2_in = m_dot_CO2;
  m_dot_CO2_out = -CO2Outlet.m_dot;
  T_CO2_in = CO2Inlet.T;
  T_CO2_out = CO2Outlet.T;
```

```
  T_CO2_out = CO2OutletTemperature;
  T_s_out = ParticleOutletTemperature;
//Mass Balance
  m_dot_s_in = m_dot_s_out;
  m_dot_s_in = rho_s * phi_s * v_s * hc_s * N_plate * W;
  m_dot_CO2_in = m_dot_CO2_out;
  m_dot_CO2_in = rho_CO2 * v_CO2 * hc_CO2 * N_plate * W;
//Energy Balance
  T_s[1] = T_s_in;
  for i in 2:n - 1 loop
    cp_s * rho_s * phi_s * der(T_s[i]) = (-cp_s * rho_s * phi_s
* v_s * (T_s[i] - T_s[i - 1]) / dx) + 2 * h_conv_sw / hc_s *
(T_m[i] - T_s[i]);
  end for;
  T_s[n] = T_s[n - 1];
  T_s[n] = T_s_out;
  T_CO2[1] = T_CO2_out;
  T_CO2[1] = T_CO2[2];
  for i in 2:n - 1 loop
    cp_CO2 * rho_CO2 * der(T_CO2[i]) = (-cp_CO2 * rho_CO2 *
v_CO2 * (T_CO2[i] - T_CO2[i + 1]) / dx) + 2 * h_conv_CO2 /
hc_CO2 * (T_m[i] - T_CO2[i]);
  end for;
  T_CO2[n] = T_CO2_in;
  T_m[1] = T_m[2];
  for i in 2:n - 1 loop
    cp_m * rho_m * der(T_m[i]) = 2 * h_conv_CO2 / t_m *
(T_CO2[i] - T_m[i]) + 2 * h_conv_sw / t_m * (T_s[i] - T_m[i]);
  end for;
  T_m[n] = T_m[n - 1];
  Q_CO2 = cp_CO2 * m_dot_CO2_in * (T_CO2_out - T_CO2_in);
//Properties
  for i in 1:n loop
    h_s[i] = Media.Particle.Enthalpy(T_s[i]);
    h_CO2[i] = Media.CO2.Enthalpy(T_CO2[i]);
  end for;
  cp_s = Media.Particle.SpecificHeat(T_s_out);
  rho_s = Media.Particle.Density();
  cp_CO2 = Media.CO2.SpecificHeat();
  rho_CO2 = Media.CO2.Density();
  cp_m = Media.StainlessSteel.SpecificHeat();
  rho_m = Media.StainlessSteel.Density();
end ParticleHeatExchanger;
```