```
model BucketElevator
  extends Icons.BucketElevator;
  // Geometry
  parameter Integer n = 100 "number of discretizations";
  parameter Types.Temperature T_o = 20 + 273.15 "ambient
temperature";
  parameter Types.Length w = 1.0 "elevator case width";
  parameter Types.Length l = 2.0 "elevator case length";
  parameter Types.Length h = 50.0 "elevator case height";
  parameter Types.Length t_m = 0.006 "elevator case thickness";
  parameter Types.Area A_s = (2 * w + 2 * l) * h "elevator case
surface area";
  parameter Types.Area A_lift = 0.05 * 0.05 "bucket surface
area";
  parameter Types.Area P_lift = 2 * 0.05 + 2 * 0.05 "bucket
perimeter";
  parameter Types.Convection h_conv_sw = 180 "solid-wall
convection coefficient";
  parameter Types.Convection h_loss = 10 "elevator heat loss
coefficient";
  parameter Real dx = h / (n - 2) "discretization size";
  // Particle Variables
  Types.MassFlowRate m_dot_s_in "inlet mass flow rate";
  Types.MassFlowRate m_dot_s_out "outlet mass flow rate";
  Types.Temperature T_s_in "inlet temperature";
  Types.Temperature T_s_out "outlet temperature";
  Types.Temperature T_s[n](each start = 273.15 + 25) "solid
temperature distribution";
  Types.Enthalpy h_s[n] "particle enthalpy";
  Types.SpecificHeatCapacity cp_s "particle specific heat";
  Types.Density rho_s "particle density";
  parameter Real phi_s = 0.6 "solid volume fraction";
  Types.Velocity v_s "solid velocity";
  Types.Heat Q_loss "heat loss";
  Types.Temperature T_m[n](each start = 273.15 + 25) "metal
temperature distribution";
  Types.SpecificHeatCapacity cp_m "metal specific heat";
  Types.Density rho_m "metal density";
  FallingParticleReceiverSystem.Interfaces.ParticleFlow
ParticleOutlet annotation(
    Placement(visible = true, transformation(origin = {0, -100},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {-60, 56}, extent = {{-10, -10},
{10, 10}}, rotation = 0)));
  FallingParticleReceiverSystem.Interfaces.ParticleFlow
ParticleInlet annotation(
```

```
    Placement(visible = true, transformation(origin = {0, 98},
extent = {{-10, -10}, {10, 10}}, rotation = 0),
iconTransformation(origin = {-60, -58}, extent = {{-10, -10},
{10, 10}}, rotation = 0)));
equation
// Connection
  m_dot_s_in = ParticleInlet.m_dot;
  m_dot_s_out = -ParticleOutlet.m_dot;
  T_s_out = ParticleOutlet.T;
  T_s_in = ParticleInlet.T;
// Mass Balance
  m_dot_s_in = m_dot_s_out;
  m_dot_s_in = rho_s * phi_s * v_s * A_lift;
// Energy Balance
  T_s[1] = T_s_in;
  for i in 2:n - 1 loop
    cp_s * rho_s * phi_s * der(T_s[i]) = (-cp_s * rho_s * phi_s
* v_s * (T_s[i] - T_s[i - 1]) / dx) + h_conv_sw / P_lift *
(T_m[i] - T_s[i]);
  end for;
  T_s[n] = T_s[n - 1];
  T_s[n] = T_s_out;
  T_m[1] = T_m[2];
  for i in 2:n - 1 loop
    cp_m * rho_m * der(T_m[i]) = h_loss / t_m * (T_o - T_m[i]) +
P_lift * h / A_s * h_conv_sw / t_m * (T_s[i] - T_m[i]);
  end for;
  T_m[n] = T_m[n - 1];
  Q_loss = sum(h_loss * (T_m[i] - T_o) * dx for i in 2:n - 1) *
(A_s / h);
// Properties
  for i in 1:n loop
    h_s[i] = Media.Particle.Enthalpy(T_s[i]);
  end for;
  cp_s = Media.Particle.SpecificHeat(T_s_out);
  rho_s = Media.Particle.Density();
  cp_m = Media.StainlessSteel.SpecificHeat();
  rho_m = Media.StainlessSteel.Density();
  annotation(
    Icon(coordinateSystem(preserveAspectRatio = false)),
    Diagram(coordinateSystem(preserveAspectRatio = false)),
    __OpenModelica_commandLineOptions = "");
end BucketElevator;
```